


Darstellung von Strukturen

Diese Anleitung umfasst verschiedene Varianten zur Darstellung von Kristallstrukturen mit Jmol: Einerseits Skriptbefehle und das Kontextmenü, die auf jeder Jmol-Webseite im Internet funktionieren und andererseits die Tools auf der Seite www.molek.ch, welche auf genau denselben Skriptbefehlen beruhen, aber manchmal schneller und bequemer sind.



Schreibt man die Skriptbefehle in eine Text-Datei, z.B. skript.spt, so kann man diese Datei per drag and drop über den Jmol-Anzeigebereich ziehen, um alle Befehle ausführen zu lassen.


Ein Hinweis zu den Auswahlmenüs: Wenn eine bestimmte Auswahl bereits einmal getroffen wurde, muss zuerst etwas anderes gewählt werden, damit die gewünschte Auswahl wieder funktioniert.

Laden einer cif-Datei

Die gewünschte .cif-Datei wird mittels drag and drop in den Anzeigebereich von jmol verschoben oder mittels Kontextmenü geöffnet (Das Kontextmenü erscheint bei einem Rechtsklick auf den Anzeigebereich, nun wählt man *Datei / Laden / ausgewählte Datei* öffnen).


Es erscheint ein erstes Bild der Struktur. Das Bild kann

- mit Linksdrag gedreht,
- mit ctrl + alt + Linksdrag verschoben
- mit alt + Linksdrag in der Bildebene gedreht werden
- Fährt man mit der Maus über ein Atom, so erscheint das Atomsymbol
- Mit Rechtsklick (Kontextmenü) / Konsole kann man nun eine Konsole öffnen. Klickt man daraufhin auf ein Atom, so erscheinen die Bezeichnung des Atoms, seine Nummer und Koordinaten



Skriptbefehle werden in den unteren Teil der Konsole geschrieben oder in die Kommandozeilen in einigen Registern, z.B. im Register *Wählen*.

Auf molek.ch kann man mit folgenden Icons den Anzeigebereich verändern und sich einen Überblick über die Mausbefehle verschaffen (auch Mac-Maus).



Wenn der Text links und das Menü unten ausgeblendet werden, hat man immer noch unten rechts eine Kommandozeile zur Verfügung.

Skriptbefehle

Eine ausführliche (wenn auch nicht vollständige) Skript-Dokumentation erhält man, indem man in der Konsole oben links «help» anklickt.

Zwischenspeichern


In Jmol kann man ausgeführte Befehle leider nur unter gewissen Umständen rückgängig machen, im Allgemeinen geht das nicht. Man muss die Arbeit also regelmässig zwischenspeichern.

Am sichersten exportiert man die Arbeit als Datei (*Datei/Speichern/alles als Jmol-Datei speichern (zip)*) oder mit dem Konsolenbefehl

```
write dateiname.jmol
write dateiname.png as pngj
```

Letzterer Befehl speichert ein Bild im png-Format, die sich z.B. in Word einfügen lässt, aber die Datei enthält zudem alle Informationen zu dem Modell – wird die Datei in Jmol gezogen, erhält man also wieder das ursprüngliche 3D-Modell. Ohne den Zusatz *as pngj* wird ein png-Bild ohne Zusatzinformationen zum Modell gespeichert.

Auf molek.ch gibt es zudem die Möglichkeit, einen Zustand zwischenzuspeichern (drei verschiedene Zustände S1, S2 und S3) und später wiederherzustellen (R1, R2, R3):



Bindungen verändern

Die Beispieldatei führt zu einer seltsamen Bindung zwischen den Aluminiumionen, da diese in der .cif-Datei nicht als Ionen, sondern als Atome gespeichert sind. Im Folgenden wird diese Bindung gelöscht und andere Bindungen werden modifiziert.

Komplexbindungen löschen und Aussehen verändern

Skriptbefehle werden in den unteren Teil der Konsole oder in eine der Kommandozeilen eingetippt oder hier kopiert und dort eingefügt:

```
connect (_Al) (_Al) delete
```


connect ist der Befehl zum Modifizieren von Bindungen, mit _Al (unterstrich-Elementsymbol) werden alle Aluminiumatome symbolisiert.

Manchmal reicht es auch nur,

```
connect
```

ezzutippen. Jmol überprüft dann alle Bindungen und verändert sie gemäss Standardeinstellungen. (Klappt hier aber nicht, da die Aluminiumatome nicht als Ionen gespeichert sind)

Komplexbindungen löschen und Aussehen verändern



```
connect 2 (_Al) (*) 0.5
```

Dieser Befehl Verbindet alle Atome (*), die maximal 0.2 nm vom Aluminiumatom (_A) entfernt sind (2, Angabe in Ångström) mit einer «halben Bindung» (0.5), also gestrichelt. Mit folgendem Befehl wird aus der halben Bindung wieder eine ganze, nun aber mit kleinerem Radius

```
connect 2 (_Al) (*) 1 radius 0.05
```


Wenn die Atome noch nicht verknüpft wären, müsste man zunächst die Spezifizierung für den Radius noch weglassen, damit der Befehl funktioniert:

```
connect 2 (_Al) (*) 1;
```

```
connect 2 (_Al) (*) 1 radius 0.05
```

Doppelbindungen einzeichnen

Fahre mit der Maus über ein O-Atom, zu dem eine Doppelbindung gezeichnet werden soll.



In einem gelben Feld erscheint die Nummer des Atoms, im vorliegenden Fall Nummer 10. Mit folgendem Code wird zwischen dem Atom Nummer 10 (@10) und allen anderen Atomen, die näher als 0.21 nm sind, eine Doppelbindung gezeichnet

```
connect 2.1 (@10) (*) 2
```

Will man für alle gleichartigen O-Atom mit einem Befehl Doppelbindungen einzeichnen, so wählt man zuerst alle O-Atome aus, die an ein C-Atom gebunden sind aber nicht an ein Aluminium-Atom (Symbol ! bedeutet «nicht»), und danach zieht man Doppelbindungen von den ausgewählten Atomen (selected) zu allen anderen (*), die näher als 2.0 nm sind (da hier mehrere Befehle aufeinander folgen, setze ich am Zeilenende jeweils einen Strichpunkt. Man kann beide Befehle gemeinsam kopieren, in die Konsole unten einfügen und mit enter ausführen lassen).

```
select (_O) and connected(_C) and  
!connected(_Al);  
connect 2.0 (selected) (*) 2;
```

Dicke der Bindungen verändern


Alle Atome auswählen oder alle ausser die Aluminium-Atome auswählen (deren Bindungen haben wir ja schon modifiziert:

```
select *;
```


```
select !_Al;  
wireframe 0.1;
```

Doppelbindungen einzeichnen

Wählen/Atom (um Atome durch Anklicken auszuwählen) und nichts (um alle Atome abzuwählen)







Klicke nun alle Atome an, die Doppelbindungen auftreten und wähle danach





Hinweis: Wenn im Auswahlfeld *Doppelbindung* bereits ausgewählt ist, so funktioniert die Wahl erst wieder, wenn man dazwischen etwas anderes ausgewählt hat.

Dicke der Bindungen verändern



	<p>Bindungsdurchmesser mit den Optionen im folgenden Bereich anpassen:</p> 
<p>Größe der Atome verändern</p> <p>Will man die Atome der Wassermoleküle gross darstellen, so wählt man sie zuerst aus. Beispielsweise kann man zuerst alle O-Atome, die mit zwei H-Atomen verbunden sind, auswählen und anschliessend auch noch die daran gebundenen Wasserstoffatome. Danach verändert man ihre Größe mit <i>spacefill</i>:</p> <pre>select _O and connected(2,_H); select selected or _H and connected(selected); spacefill 80%;</pre>	<p>Wählen/Atome/nichts und dann die Atome der Wassermoleküle anklicken und</p> 
<p>Gelbe Auswahl-Halos ausschalten</p> <pre>selectionHalos off</pre>	

Mit Bild exportieren (im Kontext-Menü, siehe unten links oder in der Konsole wie oben beschrieben mit dem Befehl write) sollte man jetzt das rechte Bild erhalten:



Layout-Optionen ausprobieren

Am besten probiert man hier ein paar Layoutoptionen aus und schaut in der Konsole, welche Befehle unterlegt sind. Um alle ausgeführten Befehle zu sehen, klickt man in der Konsole unten auf *history*.


Antialiasing bewirkt ein sehr viel schöneres Bild, die Darstellung wird allerdings dadurch viel langsamer. Diese Option hat keinen Einfluss auf die Qualität der exportierten Bilder.

Kristalle

Wir möchten nun herausfinden, ob sich die beiden Wassermoleküle an Wasserstoffbrücken beteiligen. Dazu müssen wir aber viel mehr Teilchen in der Umgebung dieser Wassermoleküle einzeichnen. Zu diesem Zweck wechseln wir ins Register *Kristall*, lassen die Ionen einer Unitcell und die Achsen einzeichnen


Folgender Befehl funktioniert nur, wenn man die Datei mit dem Kontextmenü/Datei geladen hat (drag and drop geht nicht).

```
load ""{1 1 1};
```




Wir erweitern den Kristall in Richtung der a- und c-Achse (dritte Achse), um nach weiteren Wasserstoffbrücken zu suchen.

```
load ""{2 1 3};
```




Nun werden Unitcell und die Achsen wieder ausgeblendet

```
unitcell off;  
axes off;
```




Löschen von Teilchen


Mit dem Menü kann man Teilchen löschen, indem man die zu löschenden Teilchen mit einer der folgenden Optionen anwählt (rot) und danach endgültig löscht (grün):



Wenn man mit shift+drag auswählt, aber nicht alle Atome der zu löschenden Teilchen erwischt, kann man die Auswahl auf die ganzen Teilchen ausweiten:



Alternativ kann man auch diejenigen Teilchen auswählen, die man behalten will, und dann den Rest löschen:



Angeschnittene Teilchen mit Skriptbefehlen entfernen

Gegebenenfalls muss man nun noch angeschnittene Teilchen entfernen: Für eine Reihe von mehratomigen Ionen gibt es im Register *Kristall* rechts Buttons, mit denen man angeschnittene Teilchen entfernen kann, ebenso ganze Wassermoleküle und einzelne O-, H- Atome und OH- Gruppen.

Die angeschnittenen Teilchen in diesem Beispiel fallen leider nicht in diese Kategorien. Man muss sie also gezielt anwählen und dann löschen.

Man kann zunächst alle Aluminium-Ionen wählen, die *nicht* 6 koordinative Bindungen eingehen. Diese Aluminiumionen befinden sich in angeschnittenen Gruppen (sonst würden sie ja 6 Bindungen eingehen):

```
select _Al and !connected(6)
```

Nun weitet man die Auswahl auf die ganzen Teilchen aus (also auf alle Atome innerhalb desselben „Moleküls“, wobei ein Molekül einfach alle durch Stäbe verbundenen Atome sind) und löscht:


```
select within(molecule, selected);
delete selected;
```

*Strukturen mit Hilfe von Smiles anwählen

Mittels Skript könnte man auch direkt nach den SMILES der vollständigen Teilchen suchen, dann alle mit der Auswahl verknüpften H-Atome auch noch anzuwählen, um schliesslich alles nicht Angewählte zu löschen:

```
select
  search("Al1234567.Al89%10%11%12%137.O91.N%14%15%11CCN%162CC(O)O4.C%1
  5C(O)O%10.O8C(O)C%14.O%13.O5C(O)C%16.O3.O%126");
select selected or connected selected;
delete !selected;
```

Nur: wie findet man die Smiles einer komplexen Struktur? Mit dem Menü kann man sich SMILES anzeigen lassen, indem man die fragliche Struktur auswählt und dann auf folgende Option klickt:



Um dasselbe mit Skriptbefehlen zu erreichen, stellt man zunächst im Kontext-Menü den Auswahl-Modus auf *Molekül selektionieren* oder erreicht dasselbe mit dem Skriptbefehl *set picking select MOLECULE*. Dann klickt man auf ein vollständiges Teilchen und tippt anschliessend in die Konsole:


```
show smiles
```

Den Auswahlmodus kann man im Kontextmenü oder mit dem Skriptbefehl *set picking off* wieder zurücksetzen.


Wasserstoffbrücken suchen und einzeichnen

Im Register Layout kann man mit dem Befehl Stereo in die Stereoansicht wechseln (oder im Kontextmenü: stil/stereographie/3D Brille rot-cyan), und mit Hilfe einer Stereobrille Wasserstoffbrücken suchen.

Zudem kann man Wasserstoffbrücken automatisch einzeichnen lassen.

<pre>calculate hbonds; color hbonds yellow; hbonds 0.05</pre>	
---	--

Falls das Programm nicht alle Wasserstoffbrücken automatisch erkennt, z.B. zwischen einem Chloratom und benachbarten Wasserstoffatomen, kann man zusätzliche Wasserstoffbrücken einzeichnen:

<pre>connect 3 (_Cl) (_H) hbond; color hbonds yellow; hbonds 0.05</pre>	<p>Atome auswählen, zwischen denen eine Wasserstoffbrücke gezeichnet werden soll (Wählen / Atom / nichts) und dann folgendes Pop-upmenü öffnen</p>  <p>Hier können zwischen den ausgewählten Atomen Wasserstoffbrücken eingezeichnet werden. Danach können wiederholt neue Atome ausgewählt und Wasserstoffbrücken eingezeichnet werden.</p>
---	--


Flächen zeichnen

Um bestimmte Atome oder Atomgruppen herum lassen sich Oberflächen zeichnen, indem man die betreffenden Atome zuerst anwählt und dann folgendermassen vorgeht:

Menübefehle

Im Kontextmenü (Rechtsklick) kann man mit der Option *Oberflächen* eine Reihe verschiedener Oberflächen einzeichnen. Die Oberflächen erscheinen dann oft an gewissen Stellen zerrissen, weil die Oberfläche zwischen den gewählten und den nicht gewählten Atomen unterbrochen wird.

Mit dem Menü auf molek.ch ist dieses Problem mit den Befehlen *Oberfläche 2* oder *Oberfläche Netz 2* gelöst. Allerdings sind auch hier die Möglichkeiten eingeschränkt, z.B. ist der Radius der Oberflächen auf den Van der Waals-Radius beschränkt. Viel mehr Optionen erhält man mit Skriptbefehlen (siehe unten)



Skriptbefehle

Mit Skriptbefehlen kann man auf folgende Weise Oberflächen einzeichnen:

```

isosurface of01 ignore (!selected) vdw; #1
delay 1.0; #2
isosurface of01 ignore (!selected) resolution 3 vdw 1.5 mesh nofill
  frontonly; #3
color $of01 [xff0000] translucent 0.3; #4
delay 1.0;
color $of01 green translucent 0.5; #5
delay 1.0;
isosurface of01 delete #6
delete $of01 #Alternative
delay 1.0;
isosurface delete; #7

```

- #1 Hier wird eine Oberfläche mit dem Namen of01 generiert. (diesen Namen kann man frei wählen). Will man eine Oberfläche belassen und zusätzlich eine weitere einzeichnen, muss man der neuen Oberfläche einen anderen Namen geben, z.B. of02, damit die alte nicht gelöscht und ersetzt wird.

ignore (!selected) verhindert, dass die Oberfläche aufgerissen dargestellt wird (siehe oben), indem Oberfläche gezogen wird, als wären die nicht angewählten Atome nicht erhalten (die nicht angewählten Atome werden ignoriert).

vdw oder *vanderwaals* bewirkt, dass das Netz oder die Oberfläche beim Van der Waals-Radius der Atome gezogen wird. Anstelle von *vanderwaals* wäre auch *ionic* (Ionen-Oberfläche), *solvent* (lösemittelzugängliche Oberfläche), *sasurface* (Die Oberfläche ist an der stelle, an der sich das Zentrum einer Kugel befindet, die über die Oberfläche rollt. Mit einer nachgesetzten Zahl gibt man den Radius dieser Kugel an).

- #2 Werden alle diese Befehle gleichzeitig eingegeben, so wartet jmol an dieser Stelle 1.0 Sekunden, bis der nächste Befehl ausgeführt wird.

- #3 Die Oberfläche kann mit vielen zusätzlichen mehr Optionen gezeichnet werden. Einige davon sind hier umgesetzt:

Resolution bestimmt die Anzahl Punkte auf der Oberfläche, die berechnet und miteinander verbunden werden. Standardwert ist 2.

Die Zahl nach *vanderwaals* gibt den Vergrößerungsfaktor für den Radius an (um welchen Faktor die Oberfläche gegenüber dem Van der Waals -Radius vergrößert wird) und muss zwingend eine Nachkommastelle aufweisen (oder dann um 100 sein, ohne Nachkommastelle interpretiert Jmol die Zahl als Prozent des Van der Waals-Radius).

mesh nofill bewirkt, dass ein Netz und keine Oberfläche gezogen wird. *Fill nomesh* würde das Gegenteil bewirken (und ist Standardeinstellung)

frontonly bewirkt, dass nur die Vorderseite des Netzes angezeigt wird

- #4 Dieser Befehl färbt die bezeichnete Fläche (in diesem Fall oberflaeche01) in der RGB-Farbe ff0000 (je zwei Stellen beziffern den Farbwert für rot, grün und blau, wobei die Zahlen im hexadezimalsystem notiert werden mit f als grösster Ziffer für 15. ff0000 bedeutet also: der Wert für rot ist $15 \cdot 16 + 15 = 255$, die Werte für Grün und blau sind je null). Daneben kann man die Farbe auch als Begriff eingeben, vgl. #4

Translucent bestimmt die Durchsichtigkeit mit Werten von 0 (undurchsichtig) bis 1.0 (mit Nachkommastelle: unsichtbar, ohne Nachkommastelle als Prozent interpretiert)

- #5 siehe #4. Immer, wenn man den Namen der Oberfläche (of01) verwendet ohne vorangestelltes *isosurface*, muss ein Dollar-Zeichen \$ unmittelbar vor den Namen gesetzt werden.
- #6 Nur gerade die bezeichnete Fläche wird gelöscht
- #7 Alle Flächen werden gelöscht. Wenn man also andere Flächen unter anderem Namen erzeugt hat, werden diese auch alle gelöscht.

Der Befehl *isosurface* bietet aber weit mehr Optionen. Beispielsweise können Molekülorbitale dargestellt werden oder mit dem Befehl *MAP* irgendwelche Eigenschaften der Atome auf die Oberfläche projiziert und so farblich dargestellt werden, z.B. Ladungen und Partialladungen (molecular electrostatic potential, MEP)